NAG Library Routine Document

F07BSF (ZGBTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07BSF (ZGBTRS) solves a complex band system of linear equations with multiple right-hand sides,

$$AX = B, \quad A^{\mathrm{T}}X = B \quad \text{or} \quad A^{\mathrm{H}}X = B,$$

where A has been factorized by F07BRF (ZGBTRF).

2 Specification

```
SUBROUTINE F07BSF (TRANS, N, KL, KU, NRHS, AB, LDAB, IPIV, B, LDB, INFO)

INTEGER               N, KL, KU, NRHS, LDAB, IPIV(*), LDB, INFO
COMPLEX (KIND=nag_wp) AB(LDAB,*), B(LDB,*)
CHARACTER(1)          TRANS
```

The routine may be called by its LAPACK name ***zgbtrs***.

3 Description

F07BSF (ZGBTRS) is used to solve a complex band system of linear equations $AX = B$, $A^{\mathrm{T}}X = B$ or $A^{\mathrm{H}}X = B$, the routine must be preceded by a call to F07BRF (ZGBTRF) which computes the LU factorization of A as $A = PLU$. The solution is computed by forward and backward substitution.

If TRANS = 'N', the solution is computed by solving $PLY = B$ and then $UX = Y$.

If TRANS = 'T', the solution is computed by solving $U^{\mathrm{T}}Y = B$ and then $L^{\mathrm{T}}P^{\mathrm{T}}X = Y$.

If TRANS = 'C', the solution is computed by solving $U^{\mathrm{H}}Y = B$ and then $L^{\mathrm{H}}P^{\mathrm{T}}X = Y$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: TRANS – CHARACTER(1) *Input*

On entry: indicates the form of the equations.

TRANS = 'N'
$AX = B$ is solved for X.

TRANS = 'T'
$A^{\mathrm{T}}X = B$ is solved for X.

TRANS = 'C'
$A^{\mathrm{H}}X = B$ is solved for X.

Constraint: TRANS = 'N', 'T' or 'C'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: KL – INTEGER *Input*

On entry: k_l, the number of subdiagonals within the band of the matrix A.

Constraint: $\mathrm{KL} \geq 0$.

4: KU – INTEGER *Input*

On entry: k_u, the number of superdiagonals within the band of the matrix A.

Constraint: $\mathrm{KU} \geq 0$.

5: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: $\mathrm{NRHS} \geq 0$.

6: AB(LDAB,$*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the LU factorization of A, as returned by F07BRF (ZGBTRF).

7: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BSF (ZGBTRS) is called.

Constraint: $\mathrm{LDAB} \geq 2 \times \mathrm{KL} + \mathrm{KU} + 1$.

8: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: the pivot indices, as returned by F07BRF (ZGBTRF).

9: B(LDB,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

10: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07BSF (ZGBTRS) is called.

Constraint: $\mathrm{LDB} \geq \max(1, \mathrm{N})$.

11: INFO – INTEGER *Output*

On exit: $\mathrm{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A+E)x = b$, where

$$|E| \leq c(k)\epsilon|L||U|,$$

$c(k)$ is a modest linear function of $k = k_l + k_u + 1$, and ϵ is the ***machine precision***. This assumes $k \ll n$.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(k)\,\mathrm{cond}(A,x)\epsilon$$

where $\mathrm{cond}(A,x) = \left\||A^{-1}||A||x|\right\|_\infty / \|x\|_\infty \leq \mathrm{cond}(A) = \left\||A^{-1}||A|\right\|_\infty \leq \kappa_\infty(A)$.

Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$, and $\mathrm{cond}\left(A^{\mathrm{H}}\right)$ (which is the same as $\mathrm{cond}\left(A^{\mathrm{T}}\right)$) can be much larger (or smaller) than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07BVF (ZGBRFS), and an estimate for $\kappa_\infty(A)$ can be obtained by calling F07BUF (ZGBCON) with NORM = 'I'.

8 Further Comments

The total number of real floating point operations is approximately $8n(2k_l + k_u)r$, assuming $n \gg k_l$ and $n \gg k_u$.

This routine may be followed by a call to F07BVF (ZGBRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07BEF (DGBTRS).

9 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.65+2.26i & -2.05-0.85i & 0.97-2.84i & 0.00+0.00i \\ 0.00+6.30i & -1.48-1.75i & -3.99+4.01i & 0.59-0.48i \\ 0.00+0.00i & -0.77+2.83i & -1.06+1.94i & 3.33-1.04i \\ 0.00+0.00i & 0.00+0.00i & 4.48-1.09i & -0.46-1.72i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -1.06+21.50i & 12.85+\ \ 2.84i \\ -22.72-53.90i & -70.22+21.57i \\ 28.24-38.60i & -20.7\ \ -31.23i \\ -34.56+16.73i & 26.01+31.97i \end{pmatrix}.$$

Here A is nonsymmetric and is treated as a band matrix, which must first be factorized by F07BRF (ZGBTRF).

9.1 Program Text

```
    Program f07bsfe

!     F07BSF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zgbtrf, zgbtrs
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter              :: nin = 5, nout = 6
      Character (1), Parameter        :: trans = 'N'
!     .. Local Scalars ..
      Integer                         :: i, ifail, info, j, k, kl, ku, ldab,  &
                                         ldb, n, nrhs
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:), b(:,:)
      Integer, Allocatable            :: ipiv(:)
      Character (1)                   :: clabs(1), rlabs(1)
!     .. Intrinsic Procedures ..
      Intrinsic                       :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07BSF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs, kl, ku
      ldab = 2*kl + ku + 1
      ldb = n
      Allocate (ab(ldab,n),b(ldb,nrhs),ipiv(n))

!     Read A and B from data file

      k = kl + ku + 1
      Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,n)
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Factorize A

!     The NAG name equivalent of zgbtrf is f07brf
      Call zgbtrf(n,n,kl,ku,ab,ldab,ipiv,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute solution
!       The NAG name equivalent of zgbtrs is f07bsf
        Call zgbtrs(trans,n,kl,ku,nrhs,ab,ldab,ipiv,b,ldb,info)

!       Print solution

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4', &
          'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
        Write (nout,*) 'The factor U is singular'
      End If

    End Program f07bsfe
```

9.2 Program Data

```
F07BSF Example Program Data
  4  2  1  2                                              :Values of N, NRHS, KL and KU
 (-1.65, 2.26) (-2.05,-0.85) ( 0.97,-2.84)
 ( 0.00, 6.30) (-1.48,-1.75) (-3.99, 4.01) ( 0.59,-0.48)
               (-0.77, 2.83) (-1.06, 1.94) ( 3.33,-1.04)
                             ( 4.48,-1.09) (-0.46,-1.72)  :End of matrix A
 ( -1.06, 21.50) ( 12.85,  2.84)
 (-22.72,-53.90) (-70.22, 21.57)
 ( 28.24,-38.60) (-20.73, -1.23)
 (-34.56, 16.73) ( 26.01, 31.97)                          :End of matrix B
```

9.3 Program Results

```
 F07BSF Example Program Results

 Solution(s)
                    1                  2
 1  (-3.0000, 2.0000) ( 1.0000, 6.0000)
 2  ( 1.0000,-7.0000) (-7.0000,-4.0000)
 3  (-5.0000, 4.0000) ( 3.0000, 5.0000)
 4  ( 6.0000,-8.0000) (-8.0000, 2.0000)
```